



[4.12] Random scatter plots of 'noisy' functions

The function plot feature of the TI-89/92+ will plot, at most, one point for each x-axis display pixel. If the function is rapidly changing, the resulting graph will not accurately represent the function. For example, the Y= editor was used to make this graph, with a resolution of one:



The graph shows the error for a Chebychev polynomial fit of $\ln(x)$ for $x=1$ to $x=2$. The y-axis range is about $-2.4E-10$ to $1.1E-10$. However, if we randomly choose many points on the x-axis and plot the function at each point, we get a graph that looks like this:



This type of scatter plot gives us more information about the function. Some oscillation is seen at low values of x , then the error slowly increases up to about $x = 1.75$. Beyond this point, the error suddenly increases. This graph shows at 5000 plotted points.

This program, *scatterf()*, draws this type of scatter plot.

```
scatterf(f,x,xl,xh,yl,yh,n)
Prgm
@("f(var)","var",xmin,xmax,ymin,ymax,its)
©Function scatterplot with random points
©13aug01 dburkett@infinet.com

local k,e

false→e

©Clear graph screen
clrgraph
clrdraw
plotsoff
fnoff
clrio

©Set x- & y-axis ranges
xl→xmin
xh→xmax
```

```

yl→ymin
yh→ymax

©Loop to plot the function points
for k,1,n
  if getkey()≠0:exit
  (xh-xl)*rand()+xl→xn
  expr(f&"|"&x&"="&format(xn,"S12"))→yn
  if yn<ymin or yn>ymax then
    true→e
    pxltext "range error",0,0
    disp "Range error: "&string(yn)
  else
    pton xn,yn
  endif
endfor

©Delete global variables
delvar xn,yn

©Display range errors on prgm I/O screen
if e:disp

EndPrgm

```

The input arguments are

f	The name of the function to plot, passed as a string.
x	The name of the independent variable, passed as a string.
xl	The x-axis minimum
xh	The x-axis maximum
yl	The y-axis minimum
yh	The y-axis maximum
n	The number of points to plot

A sample call would be

```
scatterf("y1(x)", "x", 1, 2, -3, 4, 1000)
```

which plots 1000 points of the function $y_4(x)$ from $x=1$ to $x=2$, with a y-axis range of -3 to 4.

scatterf() plots each point as it is calculated, so you can see the plot develop. You may stop *scatterf()* by pressing any key. Press [HOME] to return to the home screen display.

scatterf() clears the graph screen and turns off all function and data plots. Also, the program I/O screen is cleared. Note that you must specify the y-axis limits. If you set the limits too low, so that not all the points can be plotted, *scatterf()* displays the error message *range error* in the upper left screen corner, but it continues plotting. *scatterf()* also displays the y-value of that caused the range error on the program I/O screen. If range errors occurred, the program I/O screen is automatically displayed when *scatterf()* is finished, or if you press a key to stop *scatterf()*.

There is little point to using *scatterf()* for most functions. *scatterf()* is most useful when the function changes very rapidly at each x-axis pixel coordinate. For best results, thousands of points are typically be plotted.